
Construction et utilisation de l'abaque de Smith 

L'abaque de Smith est l'outil graphique le plus utilisé dans le domaine des hautes fréquences. Des 
appareils de mesure comme les analyseurs de réseau donnent une représentation de leurs 
acquisitions dans ce type d'abaque (au choix avec un plan de Bode) ; les constructeurs de 
composants HFet micro-ondes présentent également souvent les caractéristiques de leurs produits 
par le même moyen. Il s'agit donc d'un outil dont la maîtrise est indispensable pour l'étude de circuit 
haute et hyper fréquence. 
D'autre part, l'abaque de Smith donne une représentation graphique simple de phénomènes décrits 
par des équations mathématiques complexes ; elle permet une meilleure compréhension du 
comportement d'une ligne. 
Son application dépasse l'étude des phénomènes de propagation et permet en particulier de résoudre 
les problèmes d'adaptation d'impédance en général et ce qui en découle (gain, stabilité, bande 
passante et stabilité d'un amplificateur HF). 
Comme nous allons le voir, l'abaque de Smith n'est rien d'autre qu'une représentation du plan 
complexe avec une graduation permettant de lire la valeur d'impédance associée à un coefficient de 
réflexion. 
Dans la première partie nous introduirons quelques éléments nécessaires à sa construction et dans la 
seconde nous nous intéresserons aux différentes utilisations. En troisième partie, on résumera les 
principales notions à connaître sur l’abaque. 
En annexe, on trouvera les éléments nécessaires à la construction logicielle d'un abaque, avec Scilab 
(la méthode étant facilement adaptable à d'autres langages de programmation). L'intérêt de cette 
construction est essentiellement d'ordre pédagogique, de nombreux logiciels ayant été développés 
afin de permettre de travailler avec cet abaque. 
 

Eléments de construction 

Considérons une ligne sans pertes d'impédance caractéristique RC, chargée par une impédance ZU. 
La référence de longueur x=0 est prise au niveau du générateur. 

a) Coefficient de réflexion 

Rappeler l'expression du coefficient de réflexion (x)ρ  sur un point de la ligne situé à une distance x du 

générateur, en fonction des modules des tensions incidente et réfléchie, de z ainsi que de la longueur 
d'onde λ (voir exercice précédent) 
Rappeler également les propriétés du module et de la phase de (x)ρ  pour une ligne sans perte. 

b) Tension le long de la ligne 

Exprimer maintenant la valeur complexe de la tension V(x)  en un point de la ligne en fonction de la 

valeur de la tension incidente Vi (que nous prendrons comme origine des phases) et de (x)ρ  

Donner une représentation graphique dans le plan complexe et déterminer comment évolue cette 
représentation suivant que l'on se déplace sur la ligne vers le générateur ou vers la charge. 
A quoi correspond une distance de λ  sur la ligne ? 
 
Afin d'adapter la représentation à tous les cas particuliers, celle ci est normalisée par rapport à une 
tension incidente unitaire. Donner la nouvelle représentation graphique. 

c) Impédance le long de la ligne 

Rappeler l'expression du coefficient de réflexion (x)ρ  en fonction de RC et de l'impédance (x)Z  en un 

point quelconque de la ligne. 
Afin de généraliser cette expression, là aussi on normalise par rapport à RC. Exprimer alors (x)ρ  en 

fonction de la nouvelle impédance normalisée z
Z x
RC

(x) =
( )

 

 
D'après l'équation qui précède, à chaque valeur de ρ(x) , dont le module peut évoluer entre 0 et 1, 
correspond une impédance normalisée z(x). L'abaque de Smith est simplement une graduation du 
plan complexe par : 

- les lieux des points où la valeur réelle de z(x)  est constante, comme le montre la figure 
suivante : 

- les lieux des points où la valeur imaginaire de z(x)  est constante comme le montre la figure 

suivante: 

En assemblant les deux tracés précédents on obtient alors l'abaque complète : 

Afin de se familiariser avec cet abaque, repérons quelques points. Pour cela, partons de graphe de 
Fresnel établi pour les tensions aux questions précédentes et représentons les vecteurs obtenus dans 
le cas d’une impédance Z(x) infinie, puis nulle et égale à RC . Vérifier la correspondance des points 
sur l’abaque de Smith. 

En repartant de l’expression du coefficient de réflexion en fonction de l’endroit où on se trouve sur la 
ligne, déterminer comment se traduit un déplacement le long de la ligne, de la charge vers le 
générateur, puis dans le sens inverse. 
 
Que deviennent alors les points précédents lorsque l'on se déplace de λ/4 sur la ligne ? Est ce le 
résultat attendu ?  

Principe d’utilisation 

Afin d'illustrer les principales propriétés de l'abaque, nous prendrons comme exemple une ligne 
d'impédance caractéristique 50 Ω, alimentée par un générateur adapté, à une fréquence de 2 GHz. 
La longueur relative de la ligne est 1,3 λ (si on appelle λ la longueur d'onde λ=10 cm à 2 GHz avec 
une vitesse de propagation de 2 108 m/s sur la ligne). 
L'impédance de la charge vaut ZU=(25+10j)Ω. 
Les exercices suivants pourront être faits sur un abaque "papier" ou à l'aide d'un logiciel comme 
MMIP. Dans ce dernier cas, il est important afin de bien comprendre l'utilisation de l'abaque, d'avoir 
prévu sur un abaque papier, même de manière grossière, les résultats donnés par le logiciel. 

a) Représentation de l'impédance 

Nous avons vu que la graduation de l'abaque est normalisée par rapport à l'impédance 
caractéristique. 
Représenter alors le point correspondant à  zu=ZU/RC sur l'abaque. 
 
L'une des principales utilisations de l'abaque est le calcul de circuits d'adaptation d'impédance. Ceux-
ci étant composés éventuellement d'association en série et parallèle d'éléments réactifs, il est 
nécessaire de savoir réaliser cette opération d'association sur l'abaque. Comment se déplace alors le 
point correspondant à  ZU si on lui ajoute une inductance en série ? une capacité ? 

b) Lecture du coefficient de réflexion et de la tension 

Lire la valeur du coefficient de réflexion au niveau de la charge (module et phase) et vérifier par le 
calcul. 
Si l'amplitude crête du générateur vaut 5 V, lire sur l'abaque l'amplitude et la phase de la tension aux 
bornes de la charge. Vérifier par le calcul. 

c) Représentation d'un déplacement sur la ligne 

Nous avons vu la périodicité du coefficient de réflexion ρ(x)  était de λ/2, en conséquence de quoi une 

rotation complète sur l'abaque correspondait à un déplacement de λ/2 (vers la charge dans le sens 
positif, vers le générateur dans le sens négatif). 
Lire sur l'abaque l'impédance d'entrée de la ligne. Vérifier par le calcul. 
Lire également la tension à l'entrée de la ligne (en module et en phase). Vérifier par le calcul. 
Rappeler quelle tension est prise pour origine des phases. 

d) Lecture d'une admittance 

Rappeler l'expression liant le coefficient de réflexion ρ(x)  en fonction de l'impédance z(x)  en un point 
de la ligne. 

Replaçons z(x)  par son admittance y
z

(x)
(x)

=
1

 dans l'expression de ρ(x) . 

Comment sont modifiés la phase et le module de ρ(x)  ? Conclure sur la manière de lire l'admittance 
correspondant à une admittance (et réciproquement) sur l'abaque. 
Lire sur l'abaque la valeur de l'admittance à l'entrée et à la sortie de la ligne. Vérifier par le calcul. 

Suite à la remarque concernant les circuits d'adaptation d'impédance, il est également nécessaire de 
savoir ajouter une impédance en parallèle sur l'abaque. Imaginons qu'une capacité CE se trouve à 
l'entrée de la ligne, c'est à dire en parallèle sur l'impédance d'entrée ZE de celle-ci. Deux solutions 
sont possibles : 

- Après avoir déterminé le point de ZE puis de YE, d'après les calculs que nous venons de 
faire, l'ajout d'une susceptance (partie imaginaire d'une admittance) normalisée YE RC se 
traduit par un déplacement positif (la partie imaginaire de l'admittance est positive) sur un 
cercle à résistance constante. 
 
- Certains abaques, dits abaque d'impédance et d'admittance présentent deux graduations, 
une traditionnelle pour les impédances, l'autre symétrique par rapport au point (0,1) pour les 
admittances. 
La partie supérieure de cette dernière graduation correspondant aux susceptances négatives, 
on peut donc lire directement l'admittance ou l'impédance suivant la graduation utilisée. 
L'ajout d'une capacité se traduira alors par un déplacement, sur la graduation admittance, 
dans le sens négatif (la susceptance est positive mais l'échelle de l'abaque est inversée), sur 
un cercle à conductance constante. 
La figure suivante représente un abaque de Smith gradué en admittance uniquement (les 
conductances étant repérées par la lettre g et les susceptances par la lettre b) 

Déterminer par les deux méthodes la nouvelle impédance vue par le générateur si on lui ajoute une 
capacité de 1,27 pF à l'entrée de la ligne. 

e) Lecture du taux d'onde stationnaire (TOS ou ROS) 

Rappeler l'expression de la tension en un point quelconque de la ligne en fonction de la tension 
incidente Vi et du coefficient de réflexion ρ(x) . 
Lire sur l'abaque les points où on trouve les valeurs maximale VMAX et minimale VMIN de la tension 
(on supposera la ligne de longueur supérieure à λ/2). Vérifier par le calcul. 
 
On rappelle que le taux d'ondes stationnaires est défini par TOS=VMAX/VMIN. Exprimer alors ce 
dernier en fonction de ρ(x) . 

Rappeler dans le cas d'une charge résistive normalisée ru=RU/RC l'expression de celle-ci en fonction 
de ρu. 
Comparer les deux expressions obtenues et conclure sur la lecture directe du taux d'ondes 
stationnaires sur l'abaque de Smith. 
 
Lire le taux d'ondes stationnaires et vérifier par le calcul. 

f) Représentation d'un déplacement sur la ligne d'impédance caractéristique différente 

Les circuits d'adaptation d'impédances jouent couramment sur la valeur de l'impédance 
caractéristique d'une ligne. 
Imaginons que le générateur soit relié à notre ligne 50 Ω par une ligne d'impédance caractéristique 
130 Ω et de longueur relative 0,05.λ à 2 GHz. 
Pour calculer la nouvelle impédance vue par le générateur, il est nécessaire de représenter 
l'impédance d'entrée de la ligne 50 Ω (déterminée comme précédemment) dans un abaque normalisé 
130 Ω (c'est à dire de déplacer le point correspondant à ZE), puis de procéder comme précédemment 
dans le cas d'une progression sur la ligne vers le générateur. 
Calculer alors cette impédance. 

Résumé et compléments 

La maîtrise des quelques éléments que nous venons de voir permettra une utilisation correcte et 
efficace de l'abaque de Smith. Il est essentiel de retenir les points suivants : 
 

- notions de normalisation par rapport aux tensions et résistance caractéristique 

- déplacement sur l'abaque vers le générateur et vers la charge. Une rotation complète sur 

l’abaque correspond à un déplacement de λ/2 sur la ligne, vers le générateur pour le sens 
négatif et vers la charge pour le sens positif (voir convention de sens sur la première figure). 

(1,0)

déplacement
vers la charge
(sens positif)

déplacement
vers le générateur

(sens négatif)

 
 

- conversion d'une impédance en admittance et réciproquement  

(1,0)

 
- ajout d'une impédance en série et parallèle 

- courbes à coefficients de qualité constant. 

 
Cette notion est importante pour déterminer la bande passante et réponse en fréquence d’un 
quadripôle ; pour réaliser l’adaptation d’impédance d’un système à bande étroite, on peut se 
déplacer dans n’importe quel endroit de l’abaque, alors que pour un système à bande large, il faut 
veiller à rester dans une zone où le coefficient de qualité reste inférieur au rapport de la fréquence 
centrale sur la bande passante (B=f0/Q). Le coefficient de qualité est lui aussi normalisé par 
rapport à la résistance caractéristique. 

 

Q=1 

Q=2 

Q=5 

- déplacement sur une ligne d'impédance caractéristique différente de celle de la normalisation. 
Cette opération s’effectue par une normalisation du tracé par rapport à la nouvelle impédance 
caractéristique. Le déplacement le long de la ligne se fait alors de manière classique, puis on 
normalise de nouveau si nécessaire par rapport à l’ancienne impédance. 

 
Une fois ces éléments maîtrisés, on peut ensuite passer aux différentes applications et en particulier 
l'utilisation de l'abaque pour le calcul de circuits d'adaptation d'impédance. 

Annexe : programmes des illustrations 

Tracé à partie réelle d'impédance constante. 
 

clear; 
// nombre de tracés et nombre de points par tracé 
T=6; N=500; 
// valeur des résistances affichées 
param=[0   0.2   0.5   1   2    5]; 
// 
// espacement logarithmique entre les points 
abs_var=logspace(-2,3,N/2); 
var=[abs_var(N/2:-1:1),-1*abs_var]'; 
// 
// vecteur constant 
cst=ones(N,1); 
// 
// définition des résistances, réactances et impédances 
r=cst*param; 
x=var*ones(1,T); 
z=r+%i*x; 
// 
// calcul du coefficient de réflexion et vecteur tension 
un=ones(N,T); 
rau=(z-un)./(z+un); 
vect=un+rau; 
// 
// affichage 
xbasc(); 
// y=4 pour vue isométrique, z=5 pour tracer les axes 
plot2d(real(vect), imag(vect),cst',"045",nax=[2,2,2,4]); 
// 
// commentaires 
xstring( 0.03, 0, ["r=0"]); xstring( 0.2, 0, ["r=0,2"]); xstring( 0.55, 0, ["r=0,5"]);  
xstring( 0.9, 0, ["r=1"]); xstring( 1.2, 0, ["r=2"]); xstring( 1.7, 0.15, ["r=5"]); 

 
Tracé à partie imaginaire d'impédance constante. 
 

clear; 
// nombre de tracés et nombre de points par tracé 
T=6; N=500; 
// 
// valeur des réactances affichées 
param=[0   0.2   0.5   1   2    5]; 
val_x=[-1*param(T:-1:1),param]; 
// 
// espacement logarithmique entre les points 
abs_var=logspace(-2,3,N/2); 
var_r=[abs_var(N/2:-1:1) , abs_var]'; 
// 
// vecteur constant 
cst=ones(N,1); 
// 
// définition des résistances, réactances et impédances 
r=var_r*ones(1,2*T); 
x=cst*val_x; 
z=r+%i*x; 

// // 
// 
// 
// calcul du coefficient de réflexion et vecteur tension 
un=ones(N,2*T); 
rau=(z-un)./(z+un); 
vect=un+rau; 
// 
// affichage 
xbasc(); 
// y=4 pour vue isométrique, z=5 pour tracer les axes 
plot2d(real(vect), imag(vect),cst',"045",nax=[2,2,2,2]); 
// 
// commentaires 
xstring( 0.05, 0, ["x=0"]); xstring( 0.1, 0.4, ["x=0,2"]); xstring( 0.15, 0.8, ["x=0,5"]);  
xstring( 1, 1, ["x=1"]); xstring( 1.6, 0.8, ["x=2"]); xstring( 1.95, 0.4, ["x=5"]); 
xstring( 0.1, -0.45, ["x=-0,2"]); xstring( 0.15, -0.85, ["x=-0,5"]);  
xstring( 1, -1.1, ["x=-1"]); xstring( 1.6, -0.9, ["x=-2"]); xstring( 1.95, -0.45, ["x=-5"]); 

 
Tracé complet de l’abaque d’impédance. 
 

// nombre de tracés et nombre de points par tracé 
T=6; N=500; 
// 
// valeur de résistances et réactances affichées 
param_r=[0   0.2   0.5   1   2    5]; 
param_x=[-1*param_r(T:-1:1),param_r]; 
// 
// variations logarithmiques pour les résistances et réactances 
abs_var=logspace(-2,3,N/2); 
var_r=[abs_var(N/2:-1:1) , abs_var]'; 
var_x=[abs_var(N/2:-1:1),-1*abs_var]'; 
// 
// définition d'une constante 
cst=ones(N,1); 
// 
// calcul des résistances, réactances et impédances 
r1=cst*param_r; 
x1=var_x*ones(1,T); 
r2=var_r*ones(1,2*T); 
x2=cst*param_x; 
r=[r1,r2]; 
x=[x1,x2]; 
z=r+%i*x; 
// 
// calcul du coefficient de réflexion et vecteur tension 
un=ones(N,3*T); 
rau=(z-un)./(z+un); 
vect=un+rau; 
// 
// affichage  
xset("window",0);xbasc(0);  
// y=4 pour vue isométrique, z=0 pour effacer les axes 
plot2d(real(vect), imag(vect),cst',"040"); 
// 
// commentaires 
xstring( 0.03, 0, ["r=0"]); xstring( 0.2, 0, ["r=0,2"]); xstring( 0.55, 0, ["r=0,5"]);  

// 
xstring( 0.9, 0, ["r=1"]); xstring( 1.2, 0, ["r=2"]); xstring( 1.7, 0.15, ["r=5"]); 
xstring( -0.2, 0, ["x=0"]); xstring( -0.15, 0.4, ["x=0,2"]); xstring( 0.15, 0.8, ["x=0,5"]);  
xstring( 1, 1, ["x=1"]); xstring( 1.6, 0.8, ["x=2"]); xstring( 1.95, 0.4, ["x=5"]); 
// 
// 
// 
xstring( -0.2, -0.45, ["x=-0,2"]); xstring( 0.15, -0.85, ["x=-0,5"]);  
xstring( 1, -1.1, ["x=-1"]); xstring( 1.6, -0.9, ["x=-2"]); xstring( 1.95, -0.45, ["x=-5"]); 
// 
xstring(2, 0.8, ["inductance série"]); xstring(2, -0.8, ["capacité série"]); 
// 

 
Tracé de l’abaque d’admittance (à la suite du programme précédent) 
 

// 
// calcul de la nouvelle expression du coefficient de réflexion et vecteur tension 
rau_ad=-rau; 
vect_ad=un+rau_ad; 
// 
// affichage 
xset("window",1);xbasc(1); 
// y=4 pour vue isométrique, z=0 pour effacer les axes 
plot2d(real(vect_ad), imag(vect_ad),cst',"040"); 
// 
xstring( 0.25, 0.15, ["g=5"]); xstring( 0.7, 0, ["g=2"]); xstring( 1.05, 0, ["g=1"]);  
xstring( 1.35, 0, ["g=0,5"]); xstring( 1.7, -0.1, ["g=0,2"]); xstring( 1.85, 0, ["g=0"]); 
// 
xstring( 2.1, 0, ["b=0"]); xstring( -0.1, 0.4, ["b=-5"]); xstring( 0.2, 0.8, ["b=-2"]);  
xstring( 1, 1, ["b=-1"]); xstring( 1.6, 0.8, ["b=-0,5"]); xstring( 1.95, 0.4, ["b=-0,2"]); 
// 
xstring( -0.1, -0.45, ["b=5"]); xstring( 0.2, -0.85, ["b=2"]);  
xstring( 1, -1.1, ["b=1"]); xstring( 1.6, -0.9, ["b=0,5"]); xstring( 1.95, -0.45, ["b=0,2"]); 
// 
xstring(2, 0.8, ["inductance parallèle";"susceptance négative"]); 
xstring(2, -0.8, ["capacité parallèle";"susceptance positive"]); 

 
Tracé d’un déplacement sur la ligne 
 

theta=[%pi/4 : .01 : 3*%pi/2] ; 
rau_dc=.5*exp(%i*theta) ; 
vect_dc=ones(1,length(rau_dc))+rau_dc; 
xset("thickness",3); 
plot2d(real(vect_dc), imag(vect_dc),1,"000"); 

 
Tracé d’un déplacement à réactance constante 
 

x_imp=[-.6: 0.01: .6]; 
r_imp=0.3*ones(1,length(x_imp)); 
z_imp=r_imp+%i*x_imp; 
un_imp=ones(1, length(x_imp)); 
rau_imp=(z_imp-un_imp)./(z_imp+un_imp); 
vect_imp=un_imp+rau_imp; 
// 
xset("thickness",3); 

plot2d(real(vect_imp), imag(vect_imp),1,"000"); 
 
 
 
 
 
 
Tracé d’un déplacement à susceptance constante 
 

rau_adm=-rau_imp; 
vect_adm=un_imp+rau_adm; 
// 
xset("thickness",3); 
plot2d(real(vect_adm), imag(vect_adm),1,"000"); 
 
Tracé d’un déplacement à coefficient de qualité constant 
 

Q=1; 
r_q1=logspace(-4,3,N/2); 
r_q2=r_q1(N/2 : -1 : 1); 
x_q1=Q.*r_q1; 
x_q2=-Q.*r_q2; 
r_q=[r_q1,r_q2]; 
x_q=[x_q1,x_q2]; 
z_q=r_q+%i*x_q; 
rau_q=(z_q-ones(1,N))./(z_q+ones(1,N)); 
vect_q=ones(1,N)+rau_q; 
xset("thickness",3); 
plot2d(real(vect_q), imag(vect_q),1,"000"); 

Annexe 2 : abaque impédance 

Annexe 3 : abaque admittance  

